
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 7.2.3.4, which is part of the LTE test suite. 

The test case can be demonstrated to run with one LTE UE (see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
tc_7_2_3_4
Test Group:
RLC
ATS Version:
iwd-EUTRA-TVB2009-03_D09wk51
System Simulator used:
Anite CT
UE used:
LG Electronics LEO3 UE.

Verification Status:
PASS
4 Corrections required for test case 7.2.3.4
4.1 Introduction

This section describes the changes required to make test case 7.2.3.4 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-TVB2009-03_D09wk51 release.

4.2 Change 1
	Testcase name
	fl_EUTRA_7_2_3_4_TestBody

	Reason for change
	1) At step 4 and step 8, TTCN gets the current FN and Sub FN and then configures SS to transmit RLC PDU to transmit the PDU. TTCN should set the Sub FN to one of the allowed FN for TDD, otherwise SS will not be able to transmit the PDU.

2) RLC will start the t-reordering timer after it receives the PDU segment. The default value of t-reordering timer value is 80ms. In current implementation at step 14, TTCN configures SS to transmit PDU segment. Then at step 15, after 6 frames it allocates UL grant transmission. Finally after receiving the PDU at step 16, TTCN then at step 17 configures SS to transmit next PDU segment after the 10 frames. Hence the t-reordering timer expires before sending the next PDU segment. 

	Summary of change
	1) At step 4 and 8 updated Sub FN to 0 to allow PDU transmission for both FDD and TDD.
2) Updated transmission of PDU segment so that t-reordering timer does not expire at the UE.

	Source of change
	File : RLC_AM_Testcases.ttcn


Before:

	function fl_EUTRA_7_2_3_4_TestBody(inout RLC_SS_State p_RLC_Rec)

runs on EUTRA_PTC {

  var integer i;

  var RLC_DataField_Type v_RLC_Data1, v_RLC_Data2, v_RLC_Data3;

  var template (value) PDCP_DataPdu_LongSN_Type v_PDCP_DataPDU;

  var SubFrameTiming_Type v_Timing;

  //PDCP SDUs of this size will be built, appended to a PDCP header,

  //and then used as RLC SDUs

  p_RLC_Rec.TxDataSize := 21;

  // generate 12 RLC SDUs = PDCP PDUs for transmission

  for(i := 0; i < 12; i:= i + 1) {

    f_GetTxAM_PRBS (p_RLC_Rec);  // sets p_RLC_Rec.RLC_Data

    v_PDCP_DataPDU := cs_PDCP_DataPdu_LongSN(int2bit(i, tsc_PDCP_SN_Size_12 ), p_RLC_Rec.RLC_Data);

    p_RLC_Rec.RLC_DataList[i] := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(v_PDCP_DataPDU));

    p_RLC_Rec.Next_PDCP_TX_SN := (p_RLC_Rec.Next_PDCP_TX_SN + 1) mod tsc_Maximum_PDCP_SN_12;

  }

  // Step 0: The SS does not allocate any uplink grant.

  // During the whole test sequence, the SS should not allocate UL grants unless when explicitly stated so in the procedure.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  // Step 1: The SS transmits AMD PDU#1 containing a complete RLC SDU#1 without LI field.

  f_TxAMD_SDU ( p_RLC_Rec, cs_TimingInfo_Now, tsc_P_NoPoll, tsc_RLC_SDU1 );

  // Step 2: The SS transmits an uplink grant allowing the UE to transmit 1 RLC SDU.

  // Provide default grants from this point on

  f_EUTRA_StartDefULGrantTransmission ( eutra_Cell1,

                                        cs_TimingInfo_Now );

  // Step 3: Check: Does the UE transmit an AMD PDU containing RLC SDU#1?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU1 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");

  // Step 3A: Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  // Step 3B: The SS does not allocate any uplink grant.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  // Step 4: The SS transmits AMD PDU#2 containing a complete RLC SDU#2 and a complete RLC SDU#3 with one LI field.

  //100 ms to wait for having an accurate reference to start

  v_Timing := f_EUTRA_GetCurrentTiming(eutra_Cell1);

  v_Timing.SFN.Number := (v_Timing.SFN.Number + 10) mod 1024;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 0, 23);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 0, 23);

  f_TxAMD_PDU_List ( p_RLC_Rec,

                     cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                     tsc_P_NoPoll,

                     tsc_FI_FullSDU,

                     {v_RLC_Data1, v_RLC_Data2} );

  // Step 5: The SS waits for 60 ms then assigns an UL grant sufficient for the UE to loopback RLC SDU#2 and RLC SDU#3.

  v_Timing.SFN.Number := (v_Timing.SFN.Number + 6) mod 1024;

  f_EUTRA_StartDefULGrantTransmission ( eutra_Cell1,

                                        cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number ));

  // Step 6: Check: Does the UE transmit an AMD PDU containing RLC SDU#2 and RLC SDU#3 in its data field?

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 0, 23);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 0, 23);

  f_RxAMD_PDU (p_RLC_Rec,

               tsc_P_Poll,

               tsc_FI_FullSDU,

               cr_FlexPart_1LI (23),

               {v_RLC_Data1, v_RLC_Data2} );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 6");

  // Step 7: SS sends STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  // Step 7A: The SS does not allocate any uplink grant.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  // Step 8: The SS transmits AMD PDU#3 containing a complete RLC SDU#4, a complete RLC SDU#5 and a complete RLC

  // SDU#6 with two LI fields.

  v_Timing := f_EUTRA_GetCurrentTiming(eutra_Cell1);

  v_Timing.SFN.Number := (v_Timing.SFN.Number + 10) mod 1024;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4], 0, 23);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 0, 23);

  v_RLC_Data3 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], 0, 23);

  f_TxAMD_PDU_List ( p_RLC_Rec,

                     cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                     tsc_P_NoPoll,

                     tsc_FI_FullSDU,

                     {v_RLC_Data1, v_RLC_Data2, v_RLC_Data3} );

  // Step 9: The SS waits for 60 ms then assigns an UL grant sufficient for the UE to loopback RLC SDU#4, RLC SDU#5 and RLC SDU#6.

  v_Timing.SFN.Number := (v_Timing.SFN.Number + 6) mod 1024;

  f_EUTRA_StartDefULGrantTransmission ( eutra_Cell1,

                                        cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number ));

  // Step 10: Check: Does the UE transmit an AMD PDU containing RLC SDU#4, RLC SDU#5 and RLC SDU#6 in its data field?

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4], 0, 23);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 0, 23);

  v_RLC_Data3 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], 0, 23);

  f_RxAMD_PDU (p_RLC_Rec,

               tsc_P_Poll,

               tsc_FI_FullSDU,

               cr_FlexPart_2LIs (23, 23),

               {v_RLC_Data1, v_RLC_Data2, v_RLC_Data3} );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 10");

  // Step 11: SS sends STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  // Step 12: The SS does not allocate any uplink grant.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  // Step 13: The t_PollRetransmit timer for RLC PDU#5 expires and SS assumes that the transmission of AMD PDU#5

  // containing a complete RLC SDU#7, a complete RLC SDU#8, a complete RLC SDU#9, a complete RLC SDU#10, a complete

  // RLC SDU#11 and a complete RLC SDU#12 is failed and consider RLC PDU#5 for re-transmission.

  f_Delay (p_RLC_Rec.t_PollRetransmit);

  // Step 14: The SS transmits AMD PDU segment containing a complete RLC SDU#7 without LI field.

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU7], 0, 23);

  f_TxAMD_PDU_OneSeg ( p_RLC_Rec,

                       cs_TimingInfo_Now,

                       tsc_P_NoPoll,

                       tsc_FI_FullSDU,

                       tsc_LSF_NotLast,

                       0,

                       v_RLC_Data1 );

  // Step 15: The SS transmits an uplink grant allowing the UE to transmit 1 RLC SDU.

  // Provide default grants from this point on

  f_EUTRA_StartDefULGrantTransmission ( eutra_Cell1,

                                        cs_TimingInfo_Now );

  // Step 16: Check: Does the UE transmit an AMD PDU containing RLC SDU#7?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU7 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 16");

  // Step 16A: Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  // Step 16B: The SS does not allocate any uplink grant.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  // Step 17: The SS transmits AMD PDU segment containing a complete RLC SDU#8 and a complete RLC SDU#9 with one LI field.

  v_Timing := f_EUTRA_GetCurrentTiming(eutra_Cell1);

  v_Timing.SFN.Number := (v_Timing.SFN.Number + 10) mod 1024;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU8], 0, 23);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU9], 0, 23);

  f_TxAMD_SegList ( p_RLC_Rec,

                    cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                    tsc_P_NoPoll,

                    tsc_FI_FullSDU,

                    tsc_LSF_NotLast,

                    23,

                    {v_RLC_Data1, v_RLC_Data2} );

  // Step 18: The SS waits for 60 ms then assigns an UL grant sufficient for the UE to loopback RLC SDU#8 and RLC SDU#9.

  v_Timing.SFN.Number := (v_Timing.SFN.Number + 6) mod 1024;

  f_EUTRA_StartDefULGrantTransmission ( eutra_Cell1,

                                        cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number ));

  // Step 19: Check: Does the UE transmit an AMD PDU containing RLC SDU#8 and RLC SDU#9 in its data field?

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU8], 0, 23);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU9], 0, 23);

  f_RxAMD_PDU (p_RLC_Rec,

               tsc_P_Poll,

               tsc_FI_FullSDU,

               cr_FlexPart_1LI (23),

               {v_RLC_Data1, v_RLC_Data2} );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 19");

  // Step 20: SS sends STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  // Step 20A: The SS does not allocate any uplink grant.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  // Step 21: The SS transmits AMD PDU segment containing a complete RLC SDU#10, a complete RLC SDU#11 and a complete RLC

  // SDU#12 with two LI fields.

  v_Timing := f_EUTRA_GetCurrentTiming(eutra_Cell1);

  v_Timing.SFN.Number := (v_Timing.SFN.Number + 10) mod 1024;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU10], 0, 23);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU11], 0, 23);

  v_RLC_Data3 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU12], 0, 23);

  f_TxAMD_SegList ( p_RLC_Rec,

                    cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                    tsc_P_NoPoll,

                    tsc_FI_FullSDU,

                    tsc_LSF_Last,

                    69,

                    {v_RLC_Data1 , v_RLC_Data2, v_RLC_Data3} );

  // Step 22: The SS waits for 60 ms then assigns an UL grant sufficient for the UE to loopback RLC SDU#10, RLC SDU#11 and RLC SDU#12.

  v_Timing.SFN.Number := (v_Timing.SFN.Number + 6) mod 1024;

  f_EUTRA_StartDefULGrantTransmission ( eutra_Cell1,

                                        cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number ));

  // Step 23: Check: Does the UE transmit an AMD PDU containing RLC SDU#10, RLC SDU#11 and RLC SDU#12 in its data field?

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU10], 0, 23);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU11], 0, 23);

  v_RLC_Data3 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU12], 0, 23);

  f_RxAMD_PDU (p_RLC_Rec,

               tsc_P_Poll,

               tsc_FI_FullSDU,

               cr_FlexPart_2LIs (23, 23),

               {v_RLC_Data1, v_RLC_Data2, v_RLC_Data3} );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 23");

  // Step 24: SS sends STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  // Step 25: Void
} // fl_EUTRA_7_2_3_4_TestBody
	


After:

	function fl_EUTRA_7_2_3_4_TestBody(inout RLC_SS_State p_RLC_Rec)

runs on EUTRA_PTC {

  var integer i;

  var RLC_DataField_Type v_RLC_Data1, v_RLC_Data2, v_RLC_Data3;

  var template (value) PDCP_DataPdu_LongSN_Type v_PDCP_DataPDU;

  var SubFrameTiming_Type v_Timing;

  //PDCP SDUs of this size will be built, appended to a PDCP header,

  //and then used as RLC SDUs

  p_RLC_Rec.TxDataSize := 21;

  // generate 12 RLC SDUs = PDCP PDUs for transmission

  for(i := 0; i < 12; i:= i + 1) {

    f_GetTxAM_PRBS (p_RLC_Rec);  // sets p_RLC_Rec.RLC_Data

    v_PDCP_DataPDU := cs_PDCP_DataPdu_LongSN(int2bit(i, tsc_PDCP_SN_Size_12 ), p_RLC_Rec.RLC_Data);

    p_RLC_Rec.RLC_DataList[i] := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(v_PDCP_DataPDU));

    p_RLC_Rec.Next_PDCP_TX_SN := (p_RLC_Rec.Next_PDCP_TX_SN + 1) mod tsc_Maximum_PDCP_SN_12;

  }

  // Step 0: The SS does not allocate any uplink grant.

  // During the whole test sequence, the SS should not allocate UL grants unless when explicitly stated so in the procedure.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  // Step 1: The SS transmits AMD PDU#1 containing a complete RLC SDU#1 without LI field.

  f_TxAMD_SDU ( p_RLC_Rec, cs_TimingInfo_Now, tsc_P_NoPoll, tsc_RLC_SDU1 );

  // Step 2: The SS transmits an uplink grant allowing the UE to transmit 1 RLC SDU.

  // Provide default grants from this point on

  f_EUTRA_StartDefULGrantTransmission ( eutra_Cell1,

                                        cs_TimingInfo_Now );

  // Step 3: Check: Does the UE transmit an AMD PDU containing RLC SDU#1?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU1 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");

  // Step 3A: Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  // Step 3B: The SS does not allocate any uplink grant.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  // Step 4: The SS transmits AMD PDU#2 containing a complete RLC SDU#2 and a complete RLC SDU#3 with one LI field.

  //100 ms to wait for having an accurate reference to start

  v_Timing := f_EUTRA_GetCurrentTiming(eutra_Cell1);

  v_Timing.SFN.Number      := (v_Timing.SFN.Number + 10) mod 1024;

  v_Timing.Subframe.Number := 0;  \\ NOTE 1
  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 0, 23);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 0, 23);

  f_TxAMD_PDU_List ( p_RLC_Rec,

                     cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                     tsc_P_NoPoll,

                     tsc_FI_FullSDU,

                     {v_RLC_Data1, v_RLC_Data2} );

  // Step 5: The SS waits for 60 ms then assigns an UL grant sufficient for the UE to loopback RLC SDU#2 and RLC SDU#3.

  v_Timing.SFN.Number := (v_Timing.SFN.Number + 6) mod 1024;

  f_EUTRA_StartDefULGrantTransmission ( eutra_Cell1,

                                        cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number ));

  // Step 6: Check: Does the UE transmit an AMD PDU containing RLC SDU#2 and RLC SDU#3 in its data field?

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 0, 23);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 0, 23);

  f_RxAMD_PDU (p_RLC_Rec,

               tsc_P_Poll,

               tsc_FI_FullSDU,

               cr_FlexPart_1LI (23),

               {v_RLC_Data1, v_RLC_Data2} );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 6");

  // Step 7: SS sends STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  // Step 7A: The SS does not allocate any uplink grant.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  // Step 8: The SS transmits AMD PDU#3 containing a complete RLC SDU#4, a complete RLC SDU#5 and a complete RLC

  // SDU#6 with two LI fields.

  v_Timing := f_EUTRA_GetCurrentTiming(eutra_Cell1);

  v_Timing.SFN.Number      := (v_Timing.SFN.Number + 10) mod 1024;

  v_Timing.Subframe.Number := 0;  \\ NOTE 2
  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4], 0, 23);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 0, 23);

  v_RLC_Data3 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], 0, 23);

  f_TxAMD_PDU_List ( p_RLC_Rec,

                     cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                     tsc_P_NoPoll,

                     tsc_FI_FullSDU,

                     {v_RLC_Data1, v_RLC_Data2, v_RLC_Data3} );

  // Step 9: The SS waits for 60 ms then assigns an UL grant sufficient for the UE to loopback RLC SDU#4, RLC SDU#5 and RLC SDU#6.

  v_Timing.SFN.Number := (v_Timing.SFN.Number + 6) mod 1024;

  f_EUTRA_StartDefULGrantTransmission ( eutra_Cell1,

                                        cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number ));

  // Step 10: Check: Does the UE transmit an AMD PDU containing RLC SDU#4, RLC SDU#5 and RLC SDU#6 in its data field?

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4], 0, 23);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 0, 23);

  v_RLC_Data3 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], 0, 23);

  f_RxAMD_PDU (p_RLC_Rec,

               tsc_P_Poll,

               tsc_FI_FullSDU,

               cr_FlexPart_2LIs (23, 23),

               {v_RLC_Data1, v_RLC_Data2, v_RLC_Data3} );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 10");

  // Step 11: SS sends STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  // Step 12: The SS does not allocate any uplink grant.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  // Step 13: The t_PollRetransmit timer for RLC PDU#5 expires and SS assumes that the transmission of AMD PDU#5

  // containing a complete RLC SDU#7, a complete RLC SDU#8, a complete RLC SDU#9, a complete RLC SDU#10, a complete

  // RLC SDU#11 and a complete RLC SDU#12 is failed and consider RLC PDU#5 for re-transmission.

  f_Delay (p_RLC_Rec.t_PollRetransmit);

  v_Timing := f_EUTRA_GetCurrentTiming(eutra_Cell1);

  v_Timing.SFN.Number      := (v_Timing.SFN.Number + 10) mod 1024;

  v_Timing.Subframe.Number := 0;

  // Step 14: The SS transmits AMD PDU segment containing a complete RLC SDU#7 without LI field.

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU7], 0, 23);

  f_TxAMD_PDU_OneSeg ( p_RLC_Rec,

                       //cs_TimingInfo_Now,

                       cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                       tsc_P_NoPoll,

                       tsc_FI_FullSDU,

                       tsc_LSF_NotLast,

                       0,

                       v_RLC_Data1 );

//Step 17
  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU8], 0, 23);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU9], 0, 23);

  f_TxAMD_SegList ( p_RLC_Rec,

                    cs_TimingInfo(v_Timing.SFN.Number + 7, v_Timing.Subframe.Number), 

                    tsc_P_NoPoll,

                    tsc_FI_FullSDU,

                    tsc_LSF_NotLast,

                    23,

                    {v_RLC_Data1, v_RLC_Data2} );

//Step 21
  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU10], 0, 23);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU11], 0, 23);

  v_RLC_Data3 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU12], 0, 23);

  f_TxAMD_SegList ( p_RLC_Rec,

                    cs_TimingInfo(v_Timing.SFN.Number + 14, v_Timing.Subframe.Number),

                    tsc_P_NoPoll,

                    tsc_FI_FullSDU,

                    tsc_LSF_Last,

                    69,

                    {v_RLC_Data1 , v_RLC_Data2, v_RLC_Data3} );

//Step 15
  f_EUTRA_PeriodicCyclicULGrantTransmission( eutra_Cell1,

                                             cs_TimingInfo(v_Timing.SFN.Number + 6, v_Timing.Subframe.Number),

                                             7, 6, 1, 1); 

  // Step 16: Check: Does the UE transmit an AMD PDU containing RLC SDU#7?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU7 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 16");

  f_TxSTATUS_PDU(p_RLC_Rec);

//Step 18
  f_EUTRA_PeriodicCyclicULGrantTransmission( eutra_Cell1,

                                             cs_TimingInfo(v_Timing.SFN.Number + 13, v_Timing.Subframe.Number),

                                             7, 6, 1, 1); 

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU8], 0, 23);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU9], 0, 23);
//Step 19
  f_RxAMD_PDU (p_RLC_Rec,

               tsc_P_Poll,

               tsc_FI_FullSDU,

               cr_FlexPart_1LI (23),

               {v_RLC_Data1, v_RLC_Data2} );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 19");

  f_TxSTATUS_PDU(p_RLC_Rec);

//Step 22
  f_EUTRA_PeriodicCyclicULGrantTransmission( eutra_Cell1,

                                             cs_TimingInfo(v_Timing.SFN.Number + 20, v_Timing.Subframe.Number),

                                             7, 6, 1, 1);  \\ NOTE 3 
  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU10], 0, 23);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU11], 0, 23);

  v_RLC_Data3 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU12], 0, 23);
//Step23
  f_RxAMD_PDU (p_RLC_Rec,

               tsc_P_Poll,

               tsc_FI_FullSDU,

               cr_FlexPart_2LIs (23, 23),

               {v_RLC_Data1, v_RLC_Data2, v_RLC_Data3} );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 23");

  // Step 24: SS sends STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  // Step 25: Void

} // fl_EUTRA_7_2_3_4_TestBody


5 Execution Log Files

5.1 LG Electronics LEO3 UE

The LG electronics LEO3 UE passed this test case on the Anite LTE solution with Snow 3G integrity algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file tc_7_2_3_4_LG-log.html:

In the log file (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

6 References

	[1]
	R5s100019: This archive comprises text format execution log file and the TTCN file.

